
Subject: Re: "infinite" nested for
Posted by [David Fanning](#) on Mon, 25 Nov 2013 16:37:18 GMT
[View Forum Message](#) <> [Reply to Message](#)

Chriss writes:

Another way you can improve this code is to only calculate unchanging variable once, and outside the loops. For example, you have this, in which you calculate the number of elements of X, which never changes, six times!

```
for i=0, dx-1 do begin
  for j=0,dy-1 do begin
    y=reform (lm[i,j,*],N_ELEMENTS(x))
    fitErrors=FLTARR(N_ELEMENTS(x))
    T1values=FLTARR(N_ELEMENTS(x))
    S0values=FLTARR(N_ELEMENTS(x))
    fvalues=FLTARR(N_ELEMENTS(x))
    for t=0,N_ELEMENTS(x)-1 do begin
```

This can be changed to this:

```
num_x = N_Elements(x)
fitErrors=FLTARR(num_x)
T1values=FLTARR(num_x)
S0values=FLTARR(num_x)
fvalues=FLTARR(num_x)
for i=0, dx-1 do begin
  for j=0,dy-1 do begin
    y=reform (lm[i,j,*],num_x)
    for t=0,num_x-1 do begin
```

If you need to reinitialize your arrays inside the loop, then do this, rather than re-allocating more memory. I think this is probably why your program crashed after 12 hours. You ran out of dynamic memory on your machine:

```
fitErrors=Temporary(fitErrors)*0.0
T1values=Temporary(T1values)*0.0
S0values=Temporary(S0values)*0.0
fvalues=Temporary(fvalues)*0.0
```

Cheers,

David

--

David Fanning, Ph.D.
Fanning Software Consulting, Inc.

